

**sureg postestimation** — Postestimation tools for sureg

[Postestimation commands](#)   
 [predict](#)   
 [margins](#)   
 [Remarks and examples](#)  
 Also see

## Postestimation commands

The following postestimation commands are available after `sureg`:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of estimates
<code>estat ic</code>	Akaike's, consistent Akaike's, corrected Akaike's, and Schwarz's Bayesian information criteria (AIC, CAIC, AICc, and BIC)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estimates</code>	cataloging estimation results
<code>etable</code>	table of estimation results
<code>forecast</code>	dynamic forecasts and simulations
<code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of coefficients
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
<code>predict</code>	predictions and their SEs, residuals, etc.
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of estimates
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

## predict

### Description for predict

`predict` creates a new variable containing predictions such as linear predictions, standard errors, residuals, and differences between the linear predictions.

### Menu for predict

Statistics > Postestimation

### Syntax for predict

```
predict [type] newvar [if] [in] [, equation(eqno[,eqno]) statistic]
```

<i>statistic</i>	Description
------------------	-------------

<i>statistic</i>	Description
Main	
<code>xb</code>	linear prediction; the default
<code>stdp</code>	standard error of the linear prediction
<code>residuals</code>	residuals
<code>difference</code>	difference between the linear predictions of two equations
<code>stddp</code>	standard error of the difference in linear predictions

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

### Options for predict

Main

`equation(eqno[,eqno])` specifies to which equation(s) you are referring.

`equation()` is filled in with one *eqno* for the `xb`, `stdp`, and `residuals` options. `equation(#1)` would mean that the calculation is to be made for the first equation, `equation(#2)` would mean the second, and so on. You could also refer to the equations by their names. `equation(income)` would refer to the equation named `income` and `equation(hours)` to the equation named `hours`.

If you do not specify `equation()`, the results are the same as if you specified `equation(#1)`.

`difference` and `stddp` refer to between-equation concepts. To use these options, you must specify two equations, for example, `equation(#1,#2)` or `equation(income,hours)`. When two equations must be specified, `equation()` is required.

`xb`, the default, calculates the linear prediction (fitted values)—the prediction of  $\mathbf{x}_j \mathbf{b}$  for the specified equation.

`stdp` calculates the standard error of the prediction for the specified equation. It can be thought of as the standard error of the predicted expected value or mean for the observation's covariate pattern. The standard error of the prediction is also referred to as the standard error of the fitted value.

`residuals` calculates the residuals.

`difference` calculates the difference between the linear predictions of two equations in the system.

With `equation(#1,#2)`, `difference` computes the prediction of `equation(#1)` minus the prediction of `equation(#2)`.

`stdp` is allowed only after you have previously fit a multiple-equation model. The standard error of the difference in linear predictions ( $\mathbf{x}_{1j}\mathbf{b} - \mathbf{x}_{2j}\mathbf{b}$ ) between equations 1 and 2 is calculated.

For more information on using `predict` after multiple-equation estimation commands, see [R] [predict](#).

## margins

### Description for margins

`margins` estimates margins of response for linear predictions and differences between the linear predictions.

### Menu for margins

Statistics > Postestimation

### Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

<i>statistic</i>	Description
<code>default</code>	linear predictions for each equation
<code>xb</code>	linear prediction for a specified equation
<code>difference</code>	difference between the linear predictions of two equations
<code>stdp</code>	not allowed with <code>margins</code>
<code>residuals</code>	not allowed with <code>margins</code>
<code>stddp</code>	not allowed with <code>margins</code>

`xb` defaults to the first equation.

Statistics not allowed with `margins` are functions of stochastic quantities other than  $\mathbf{e}(\mathbf{b})$ .

For the full syntax, see [R] [margins](#).

## Remarks and examples

For an example of cross-equation testing of parameters using the `test` command, see [example 1](#) in [\[R\] sureg](#).

### ▷ Example 1

In [example 1](#) of [\[R\] sureg](#), we fit a seemingly unrelated regressions model of price and weight. Here we obtain the fitted values.

```
. use https://www.stata-press.com/data/r18/auto
(1978 automobile data)
. sureg (price foreign length) (weight foreign length), small dfk
(output omitted)
. predict phat, equation(price)
(option xb assumed; fitted values)
. predict what, equation(weight)
(option xb assumed; fitted values)
. summarize price phat weight what
```

Variable	Obs	Mean	Std. dev.	Min	Max
price	74	6165.257	2949.496	3291	15906
phat	74	6165.257	1656.407	1639.872	9398.138
weight	74	3019.459	777.1936	1760	4840
what	74	3019.459	736.9666	1481.199	4476.331

Just as in single-equation OLS regression, in a SURE model the sample mean of the fitted values for an equation equals the sample mean of the dependent variable.



### ▷ Example 2

Suppose that for whatever reason we were interested in the difference between the predicted values of price and weight. `predict` has an option to compute this difference in one step:

```
. predict diff, equation(price, weight) difference
```

`diff` is the same as `phat - what`:

```
. generate mydiff = phat - what
. summarize diff mydiff
```

Variable	Obs	Mean	Std. dev.	Min	Max
diff	74	3145.797	1233.26	-132.2275	5505.914
mydiff	74	3145.797	1233.26	-132.2275	5505.914



## Also see

[\[R\] sureg](#) — Zellner's seemingly unrelated regression

[\[U\] 20 Estimation and postestimation commands](#)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).